

## Appendix B: SAS Software Program

```
*** DECTOOL.SAS
*** Produces tables used in Wilde, 2000.
*** "Understanding the Food Stamp Benefit Formula"
*** Input data: qcfy1998.sd2
*** Parke Wilde
*** 9/1/2000;

*** Define names of input data sets ***;
libname out 'e:/temp/ces/';
libname in 'e:/data/qc/qc98/';

options linesize=100;

*** Create data file DECOMP ***;
data out.decomp;
set in.qc fy1998
  (keep = fsndis fsnelder fsnkid fsngmom fsusize fywgt
    fsgrinc fsnetinc fstotded fstotde2
    fscsexp fsstdded fsernded fsdepded fsmedded
    fssltded shelcap benmax fsben
    fsmnben netscrn state fssltexp
    urbrur tpov region);

*** SHELTER SUB-COMPONENT EFFECTS ***;
neldis=fsndis+fsnelder;
if neldis ge 1 then d_diseld=1;
else d_diseld=0;
if d_diseld=0 then hstar=shelcap;
else hstar=fssltexp;
n1 = max(fsgrinc-(fsstdded+fsernded+fscsexp+fsmedded+fsdepded),0);
e261 = 0.3*fssltexp;          ** gross shelter;
e262 = -0.3*min(0.5*n1,fssltexp);  ** half-income rule;
e263 = min((0.3*hstar)-(e261+e262),0); ** shelter cap effect;
e26 = e261+e262+e263;        ** shelter ded effect;

*** DEDUCTIONS EFFECTS ***;
sumded = fsstdded+fsernded+fscsexp+fsmedded+fsdepded+(e26/0.3);
e21 = 0.3*fsstdded;          ** standard;
e22 = 0.3*fsernded;          ** earned income;
e23 = 0.3*fsdepded;          ** dependent care;
e24 = 0.3*fsmedded;          ** medical;
e25 = 0.3*fscsexp;           ** child support;
e2 = e21+e22+e23+e24+e25+e26; ** deductions effect;

*** MAIN COMPONENT EFFECTS ***;
e1 = -0.3*fsgrinc;           ** income effect;
e3 = -0.3*max(0,(sumded-fsgrinc)); ** max effect;
e4 = max(0,10-(benmax-0.3*fsnetinc)); ** min effect;

*** ALTERNATIVE DEFINITIONS FOR DOUBLE-CHECKING ***;
alte26 = 0.3*fssltded;        ** alt shelt ded effect;
alte2 = 0.3*fstotded;         ** alt deductions effect;
sumfsben = benmax+e1+e2+e3+e4; ** total fs ben;

** INCOME CATEGORY ***;
If tpov le 50 then incat='1.below 50% pov';
else if 50 < tpov le 100 then incat='2.50-100% pov';
else if 100 < tpov then incat='3.above 100% pov';

*** DEMOGRAPHIC CATEGORY ***;
if d_diseld=1 and neldis ge fsusize then demcat='1.all_dis/eld';
else if d_diseld=1 then demcat='2.some_dis/eld';
else if fsngmom=1 then demcat='3.sngmom';
else if fsnkid>0 then demcat='4.other_kids';
else demcat='5.other_nokids';
```

```

*** DUMMY FOR RECEIPT OF MEDICAL ***;
IF E24>0 then yesmed=1; else yesmed=0;

*** Observations with benefits below 10 ***;
if fsben < 10 then lowflag=1; else lowflag=0;
if fsusize>2 then fsubig=1; else fsubig=0;

*** PER/PERSON BASIS ***;
* (put comments around this section for hh level analysis option) *;
hh_max = benmax;
hh_e1 = e1;
hh_e2 = e2;
hh_e3 = e3;
hh_e4 = e4;
hh_fsbm = fsben;
array effects benmax e1-e4 e21-e26 e261-e263 alte2 alte26 sumfsben fsben;
do over effects;
  effects=effects/fsusize;
end;

*** Net income of zero ***;
if n1 le 0 then netzero=1; else netzero=0;

*** Missing shelter values ***;
if (alte26 ne .) and (e261 ne .) and (alte2 ne .) and (e23 ne .);

*** Non-continental US states and territories ***;
if state=2 or state=15 or state=66 or state=72 or state=78 then outstate=1;
else outstate=0;

label
e1 = '. income effect'
e2 = '. deductions effect'
e3 = '. max effect'
e4 = '. min effect'
e21 = '. standard deduction effect'
e22 = '. earnings deduction effect'
e23 = '. dependent care ded. effect'
e24 = '. medical deduction effect'
e25 = '. child support ded. effect'
e26 = '. shelter deduction effect'
e261 = '. gross shelter exp. effect'
e262 = '. half-income rule effect'
e263 = '. shelter cap effect'
alte2 = 'DEDUCTION EFFECT (ALT.)'
alte26 = 'SHELTER DEDUCTION EFFECT (ALT.)'
sumfsben = 'COMPUTED FOOD STAMP BENEFITS (ALT.)'
d_diseld = 'DISABLED/ELDERLY PRESENT';

*** Missing values to aid in formatting output ***;
array missing miss1-miss6;
do over missing; missing=.; end;

run;

*** COUNT UNITS WITH BENEFITS BELOW $10 ***;
proc freq;
  table lowflag*fsubig;
run;

*** REMOVE OBS THAT DON'T OBEY MINIMUM BENEFIT ***;
data out.decomp;
  set out.decomp;
  if lowflag=1 then delete;
run;

*** DESCRIPTIVE FREQUENCIES ***;

```

```

proc freq data=out.decomp;
  weight fywgt;
  table fsngmom*d_diseld;
  table fsusize*demcat;
  table inccat*demcat;
  table region*demcat;
  table yesmed*demcat;
  table netzero;
run;

*** PRODUCE TABLE 1 ***;
proc means n sumwgt mean std data=out.decomp;
  where fsusize=2 and outstate=0;
  weight fywgt;
  var hh_max hh_e1 hh_e2 hh_e3 hh_e4 hh_fsben;
run;

*** ANALYSIS BY INCOME CATEGORY ***;
proc sort data=out.decomp;
  by inccat;
run;
proc means n sumwgt mean std data=out.decomp;
  weight fywgt;
  var benmax e1 e2 e3 e4 miss1 miss2
    e21 e22 e23 e24 e25 e26 miss3 miss4
    e261 e262 e263 miss5 miss6
    fsusize d_diseld alte2 alte26 sumfsben fsben;
run;
proc means n sumwgt mean std data=out.decomp;
  by inccat;
  weight fywgt;
  var benmax e1 e2 e3 e4 miss1 miss2
    e21 e22 e23 e24 e25 e26 miss3 miss4
    e261 e262 e263 miss5 miss6
    fsusize d_diseld alte2 alte26 sumfsben fsben;
run;

*** ANALYSIS BY DEMOGRAPHIC CATEGORY ***;
proc sort data=out.decomp;
  by demcat;
run;
proc means n sumwgt mean std data=out.decomp;
  weight fywgt;
  var fsusize fsnkid fsnelder fsndis;
run;
proc means n sumwgt mean std data=out.decomp;
  weight fywgt;
  by demcat;
  var fsusize fsnkid fsnelder fsndis;
run;
proc means n sumwgt mean std data=out.decomp;
  by demcat;
  weight fywgt;
  var benmax e1 e2 e3 e4 miss1 miss2
    e21 e22 e23 e24 e25 e26 miss3 miss4
    e261 e262 e263 miss5 miss6
    fsusize d_diseld alte2 alte26 sumfsben fsben;
run;

*** ANALYSIS BY HOUSEHOLD SIZE ***;
proc sort data=out.decomp;
  by fsusize;
run;
proc means n sumwgt mean std data=out.decomp;
  where fsusize le 6;
  by fsusize;
  weight fywgt;
  var benmax e1 e2 e3 e4 miss1 miss2

```

```

e21 e22 e23 e24 e25 e26 miss3 miss4
e261 e262 e263 miss5 miss6
fsusize d_diseld alte2 alte26 sumfsben fsben;
run;

*** ANALYSIS BY CENSUS REGION ***;
proc sort data=out.decomp;
by region;
run;
proc means n sumwgt mean std data=out.decomp;
by region;
weight fywgt;
var benmax e1 e2 e3 e4 miss1 miss2
e21 e22 e23 e24 e25 e26 miss3 miss4
e261 e262 e263 miss5 miss6
fsusize d_diseld alte2 alte26 sumfsben fsben;
run;

*** ANALYSIS BY RECEIPT OF MEDICAL DEDUCTION ***;
proc sort data=out.decomp;
by yesmed;
run;
proc means n sumwgt mean std data=out.decomp;
by yesmed;
weight fywgt;
var benmax e1 e2 e3 e4 miss1 miss2
e21 e22 e23 e24 e25 e26 miss3 miss4
e261 e262 e263 miss5 miss6
fsusize d_diseld alte2 alte26 sumfsben fsben;
run;

```